An Object Factory

If you have been using Visual FoxPro for any period of time you are undoubtedly very familiar with the CREATEOBJECT() function. Many developers use this function directly and liberally throughout the source code of their application. In many situations, this is perfectly acceptable; in many other situations this approach is far too inflexible. 

In order to get the CREATEOBJECT() function to work you must provide several pieces of information. First, you must pass the name of the class definition, via parameter, to the CREATEOBJECT() function. Second, the environment must be set up correctly in the sense that the program or the class library that contains the class definition must be opened by the SET PROC or SET CLASSLIB commands. Furthermore, if the object you want to instantiate is an aggregate whose parts are included in different class libraries or programs, these must be opened in the environment as well. If they are not, you end up seeing the following annoying message Class definition "name" is not found (Error 1733). I'm sure you've seen it from time to time. So, there are three parts that must be present before you can manufacture an object, the class name (class definition), the class library that contains the class name and a properly set up environment. Hum, a little more complex than it appears at first glance. There must be a way to encapsulate this complexity and make the manufacture of objects in VFP much simpler. Luckily, there is, and the creational pattern that makes this possible is known as a factory.

In the business world, factories bring together raw materials, add processes and in the end, create a product. The analogy for a software factory is the same except the raw materials are files (programs and class libraries) and Visual FoxPro commands; the process is the class I'm going to describe below and the product is a brand new baby VFP object. Put them together and you have a factory that brings together all of the raw materials required to manufacture objects.

The Raw Materials

The next problem to tackle is to determine where to store the raw materials required to manufacture the object. The only logical conclusion is a table. The table, which I named FACTORY.DBF, has a structure as follows:

Each row in this table defines the information required to manufacture an object. Notice that the table contains columns to store the class name (cClass), the class library (cLibrary) where the class name is defined and the additional class libraries (mClasses) and programs (mProcs) required to support the object if it is an aggregate. You should also take note that the FACTORY.DBF contains several other columns as well, cKey, cFile, mPreProc and finally, mPostProc. Together, these columns contain all of the raw materials required to manufacture an object.

The Class Definition (cClass)

This column contains the name of the class definition to use to create the object. At the time of object creation the value contained in this column becomes the parameter passed to the CREATEOBJECT() function.

The Class Library (cLibrary)

This column contains the name of the class library or procedure that stores the class definition. This information is used to make sure the proper class library or procedure has been opened up to support the creation of the object.

The Supporting Classes and Procedures (mClasses and mProcs)

Each of these columns contain a comma delimited list of class library or procedure files that must be opened up to support the creation and functionality of the object. Note that tests are conducted to make sure that a class library or procedure is not opened more than once.

The Object Key (cKey)

The information you store in cKey is one of the most powerful features of the Object Factory. This key allows you to assign an alias to an object that you want to create. For example, I'll assume that you are creating an accounting application that allows your customer's to create invoices. In that application is a default invoice object provided by your company whose class definition is "boInvoice". If you are not using the Object Factory or an equivalent, the code you use to instantiate your invoice could look like this:

loInvoice = CREATEOBJ('boInvoice')

Notice that you have limited your application significantly by hard coding the name of the class definition into the source. A better approach is to associate the class definition "boInvoice" with an alias. I'll use "InvoiceObject" as an example of an alias. You can now use the Object Factory to manufacture the object associated with the alias, using a wrapper function I'll illustrate later. The factory object then looks up the object definition in the FACTORY.DBF using the "InvoiceObject" key, determines the class name to pass into the CREATEOBJECT() function and the product is the object of your or your user's choice. 

Why is this better? For many reasons and the biggest of all is extensibility. Assume for an instant that you have a very large client and that client can't use your invoice form. They need an enhanced version. Wouldn't it be a nice sales technique to tell them to subclass yours, thereby inheriting all of your functionality, and make the modifications they desire. Then, to use their own newly created invoice object all they have to do is simply change the class name and class library of the "InvoiceObject" in the FACTORY.DBF from "boInvoice" to "boCustomInvoice"? Pretty cool, huh? Also notice that your initial source code is preserved via the subclass. This means that the client can take advantage of future updates from you without overwriting their own custom work.

The File (cFile) 

This column provides an enhanced support for the SET CLASSLIB command. The format for this command, straight from the VFP Help file, is as follows:

SET CLASSLIB TO ClassLibraryName [IN APPFileName | EXEFileName] [ADDITIVE] 

Notice that class libraries can also be located in other files like applications, .APP, and executables, .EXE. When the class library containing the object to be manufactured is contained in a file of this type, you must identify the name of that file. When a value is present in the cFile, the SET CLASSLIB command is expanded by inserting the value contained in that column into the IN clause of the command. Now the CREATEOBJECT() function can find a class library that is hidden in an .APP or .EXE

Pre-Processing (mPreProc) and Post-Processing (mPostProc)

As a final spit and polish for the Object Factory, I decided to throw in another feature of pre and post processing. These memo fields provide a place for you to write an entire program (not recommended) or place a call to an external function or program (recommended) that must execute before and/or after the object itself is created. 

Okay, that describes all of the parts. I'll now discuss the process of putting those parts together to manufacture an object.

The Process

The controlling method for the Object Factory is an example of a behavioral pattern called the Template Method. Technically stated Template Methods "Define the skeleton of an algorithm in an operation, deferring some steps to subclasses. Template Methods let subclasses redefine certain steps of an algorithm without changing the algorithm's structure." The process used to manufacture objects in the Object factory is as follows:

*-------- Location Section -------------

*} Program: CFactory.PRG

*} Class: CAbstractFactory

*} Method: ManufactureObject()

*---------------------------------------

LPARAMETERS tcKey, ;

roObject, ;

tlSkipEnvironmentSetup

LOCAL loObject , ;

lcClassName , ;

lcClassLibrary , ;

lcClassFile , ;

lcClassLibList , ;

lcProcList , ;

lcPreProc , ;

lcPostProc , ;

llRetVal

roObject = .NULL.

IF tlSkipEnvironmentSetup

lcClassName = tcKey

ELSE

This.GetClassInformation( tcKey , ;

@lcClassName , ;

@lcClassLibrary , ;

@lcClassFile , ;

@lcClassLibList , ;

@lcProcList , ;

@lcPreProc , ;

@lcPostProc )

ENDIF 

IF .NOT. tlSkipEnvironmentSetup

This.SetClassLibraries( lcClassLibrary , ;

lcClassFile , ;

lcClassLibList )

This.SetProcedures( lcProcList )

This.ExecuteProcedure( lcPreProc )

ENDIF

 

roObject = This.InstantiateObject( lcClassName )

IF .NOT. tlSkipEnvironmentSetup

This.ExecuteProcedure( lcPostProc )

ENDIF

 

llRetVal = TYPE('roObject.Name') == "C"

 

RETURN llRetVal

The Benefits of a Template Method

I want you to notice several things about this particular use of the Template Method pattern. The first and most important is that this method is rather English-like in appearance. It is quite apparent, if you ignore the tlSkipEnvironmentSetup switch, that the following things must occur, in this order, to manufacture an object. First it must GetClassInformation from the FACTORY table, next it must SetClassLibraries and SetProcedures that are required to support the newly created object, next it must ExecuteProcedures that are designated as pre-processing, then it executes the IntantiateObject process. Finally, it ExecutesProcedures that are designated as post-processing. The point here is that it would take a novice programmer less than 15 seconds to determine what was being accomplished here. Template Methods make maintaining code much easier for junior programmers.

Second, notice how the Template Method allows subclasses to redefine certain steps of an algorithm. Assume for a moment that you subclass the Factory Object into a new subclass named DocumentFactory. In this new subclass it is possible to override any one or all of these processes and replace them with implementation specific processing as needed. As long as you don't override the primary method, ManufactureObject(), the order of processing defined in the template method remains intact.

Using the Object Factory, Part 1 - The Starter Program, Factory.PRG

When I first started using the Object Factory in the framework, I had to create the Object Factory itself to use it. That got old after the second time. Therefore, I created a program that wraps up the complexity of using the Object Factory and named it FACTORY.PRG. The code required to use the Factory Object change from this

loInvoice = .NULL.

loFactory = CREATEOBJECT('cAbstractFactory')

IF loFactory.ManfufactureObject('InvoiceObject',;

@loInvoice) = .T.

<do some processing here>

ENDIF

to this

loInvoice = .NULL.

IF Factory('InvoiceObject',@loInvoice ) = .T.

<do some processing here>

ENDIF

This method of invoking the Object Factory is quite handy when executing it from the Command Window. Previously, in the framework, I'd have to launch the entire application just to test a form. The simple reason for this is that the form would not work unless the environment was properly set up. Now, using the Object Factory, I can test forms from the Command Window without the arduous task of starting the application first.

The listing for the FACTORY.PRG follows:

*-- Factory.PRG

LPARAMETER tcKey, roObject, tlSkipEnvironmentSetup

LOCAL lcProcList, llRetVal, loFactory

lcProcList = UPPER(SET('PROC'))

IF .NOT. ( "CFACTORY.FXP" $ lcProcList )

SET PROC TO ..\..\COMMON\LIBS\CFACTORY.FXP ADDITIVE

ENDIF

IF .NOT. ( "IFACTORY.FXP" $ lcProcList )

SET PROC TO ..\..\COMMON\ILIBS\IFACTORY.FXP ADDITIVE

ENDIF

loFactory = CREATEOBJ( 'cAbstractFactory' )

IF TYPE('loFactory.Name') == "C"

loFactory.ManufactureObject( tcKey, ;

@roObject, ;

tlSkipEnvironmentSetup )

llRetVal = TYPE( 'roObject.Name' ) == "C"

ELSE

llRetVal = .F.

ENDIF

RETURN llRetVal

Using the Object Factory, Part II - Keeping a Public Reference Around

The other option you have is to keep a reference to the Object Factory hanging around at all times. The previous example forces you to endure the object creation process two times for each object you create. The first is the creation of the Object Factory itself and the second is the object the factory eventually creates. Some of you may think this is inefficient. If so, simply add a property to your global application called .oFactory and during the instantiation process create the Object Factory and store a reference to it there. Then, each time you want to manufacture an object all you have to do is use the following signature:

<goApp>.oFactory.ManufactureObject( <cKey> , ;

@<oObject> , ;

[<lSkipEnvironmentSetup>] )

This method of use is preferred within a functioning application.

The Signature

Ah, finally, how to use the thing. The first step in this process is to populate the FACTORY.DBF with all of the object's required information. Next, create a variable into which the newly manufactured object is to be stored. Finally, invoke the Factory Object by calling its ManufactureObject method. Pass to that method the key associated with the object that you want to create in the first parameter and the variable into which the object should be stored in the second parameter (by reference). For example:

PRIVATE loInvoice

LoInvoice = .NULL.

IF <goApp>.oFactory.ManufactureObject( "InvoiceObject", ;

@loInvoice )

<work with the object as usual>

ELSE

<perform some exception processing>

ENDIF

If the method returns a .T. it means the Object Factory successfully created the desired object. A .F. means the object could not be created for some reason.

Remember, no matter which method you use to invoke the Object Factory (FACTORY.PRG or a persisting object reference), the signature remains the same. The first parameter is the key associated with the object in the FACTORY.DBF. The second parameter is the variable into which the newly created object is to be stored.

Simplifying the Object Manufacturing Process

The final modification to make to the Object Factory is to be able to use it without having to populate the FACTORY.DBF. To do this, simply pass a .T. into the ManufactureObject() method. This skips the processes where the class information is extracted out of the table and uses the value passed into the first parameter as the class name instead of the alias. For example, the following code creates an object from the "boInvoice" class definition effectively bypassing the FACTORY.DBF altogether:

LOCAL loInvoice

loInvoice = .NULL.

<goApp>.oFactory.ManufactureObject("boInvoice, @loInvoice, .T. )

Summary

Hopefully, I've convinced you that employing the use of an Object Factory to create your objects provides your application with increased flexibility and extensibility. In general, follow these rules. Rule 1: Use the FACTORY.DBF to create objects that a user may want to substitute with his own. Rule 2: Use the simplified signature of the Object Factory to completely encapsulate the object creation processing of your application. Notice that when you use this method you never have to test whether or not the object was successfully created. You know this by the return value from the ManufactureObject() method. 

As always, as more and more people adopt a methodology, the many and varied implementations of that methodology glaringly highlight any shortcomings of the initial design. If any of you run into a situation where this design falls short and should be modified please feel free to contact me and we'll begin the redesign process. Finally, comes Rule 3: In object-oriented development you should always embrace iteration.

